Specification of Source §1 Lazy—2021 edition

Jellouli Ahmed, Ian Kendall Duncan, Cruz Jomari Evangelista, Martin Henz, Alden Tan

National University of Singapore
School of Computing

April 16, 2024

The language Source is the official language of the textbook Structure and Interpretation of Computer Programs, JavaScript Adaptation. Source is a sublanguage of ECMAScript 2018 (9th Edition) and defined in the documents titled "Source §x", where x refers to the respective textbook chapter.
Source §1 Lazy is a lazy-evaluation variant of Source §1.

1 Changes

Source §1 Lazy modifies Source §1 by using lazy evaluation. In this scheme, the argument expressions of functions are passed un-evaluated to the function to which they are applied. The function then evaluates these expressions whenever their values are required. This evaluation method is generally called normal order reduction. If functions do not have any side-effects, as in Source §1, there is no need to evaluate such an expression multiple times, as the result is guaranteed to be the same. This observation leads to the variant of normal order reduction, called lazy evaluation. In the lazy evaluation language Source §1 Lazy, the evaluator remembers the result of evaluating the argument expressions for the first time, and simply retrieves this result whenever it is required again.
Lazy evaluation is used for all arguments of user-defined functions, but not for any arguments of primitive functions or operators.

2 Syntax

A Source program is a program, defined using Backus-Naur Form[1] as follows:

[1]We adopt Henry Ledgard's BNF variant that he described in A human engineered variant of BNF, ACM SIGPLAN Notices, Volume 15 Issue 10, October 1980, Pages 57-62. In our grammars, we use `bold` font for keywords, italics for syntactic variables, ϵ for nothing, $x \mid y$ for x or y, [x] for an optional x, x... for zero or more repetitions of x, and (x) for clarifying the structure of BNF expressions.

program	::=	import-directive... statement...	program
import-directive	::=	`import` { import-names } `from` string `;`	import directive
import-names	::=	ϵ | import-name (`,` import-name)...	import name list
import-name	::=	name | name `as` name	import name
statement	::=	`const` name = expression `;`	constant declaration
	|	`function` name `(` names `)` block	function declaration
	|	`return` expression `;`	return statement
	|	if-statement	conditional statement
	|	block	block statement
	|	expression `;`	expression statement
	|	`debugger` `;`	breakpoint
names	::=	ϵ | name (`,` name)...	name list
if-statement	::=	`if` `(` expression `)` block	
		`else` (block | if-statement)	conditional statement
block	::=	`{` statement... `}`	block statement
expression	::=	number	primitive number expression
	|	`true` | `false`	primitive boolean expression
	|	string	primitive string expression
	|	name	name expression
	|	expression binary-operator expression	binary operator combination
	|	unary-operator expression	unary operator combination
	|	expression binary-logical expression	logical composition
	|	expression `(` expressions `)`	function application
	|	(name | `(` names `)`) => expression	lambda expression (expr. body)
	|	(name | `(` names `)`) => block	lambda expression (block body)
	|	expression `?` expression `:` expression	conditional expression
	|	`(` expression `)`	parenthesised expression
binary-operator	::=	`+` | `-` | `*` | `/` | `%` | `===` | `!==`	
	|	`>` | `<` | `>=` | `<=`	binary operator
unary-operator	::=	`!` | `-`	unary operator
binary-logical	::=	`&&` | `||`	logical composition symbol
expressions	::=	ϵ | expression (`,` expression)...	argument expressions

Restrictions

- Return statements are only allowed in bodies of functions.
- There cannot be any newline character between `return` and expression in return statements.[2]
- There cannot be any newline character between (name | `(` parameters `)`) and `=>` in function definition expressions.[3]
- Implementations of Source are allowed to treat function declaration as syntactic sugar for constant declaration.[4] Source programmers need to make sure that functions are not called before their corresponding function declaration is evaluated.

Import directives

Import directives allow programs to import values from modules and bind them to names, whose scope is the entire program in which the import directive occurs. Import directives can only appear at the top-level. All names that appear in import directives must be distinct, and must also be distinct from all top-level variables. The Source specifications do not specify how modules are programmed.

Logical Composition

Conjunction

$$\text{expression}_1 \texttt{ \&\& } \text{expression}_2$$

stands for

$$\text{expression}_1 \texttt{ ? } \text{expression}_2 \texttt{ : false}$$

Disjunction

$$\text{expression}_1 \texttt{ || } \text{expression}_2$$

stands for

$$\text{expression}_1 \texttt{ ? true : } \text{expression}_2$$

Names

Names[5] start with _, `$` or a letter[6] and contain only _, `$`, letters or digits[7]. Restricted words[8] are not allowed as names.
Valid names are `x`, `_45`, `$$` and π, but always keep in mind that programming is communicating and that the familiarity of the audience with the characters used in names is an important aspect of program readability.

Numbers

We use decimal notation for numbers, with an optional decimal dot. "Scientific notation" (multiplying the number with 10^x) is indicated with the letter `e`, followed by the exponent x. Examples for numbers are `5432`, `-5432.109`, and `-43.21e-45`.

[2] Source inherits this syntactic quirk of JavaScript.
[3] ditto
[4] ECMAScript prescribes "hoisting" of function declarations to the beginning of the surrounding block. Programs that rely on this feature will run fine in JavaScript but might encounter a runtime error "Cannot access name before initialization" in a Source implementation.
[5] In ECMAScript 2020 (9th Edition), these names are called identifiers.
[6] By letter we mean Unicode letters (L) or letter numbers (Nl).
[7] By digit we mean characters in the Unicode categories Nd (including the decimal digits 0, 1, 2, 3, 4, 5, 6, 7, 8, 9), Mn, Mc and Pc.
[8] By restricted word we mean any of: `arguments`, `await`, `break`, `case`, `catch`, `class`, `const`, `continue`, `debugger`, `default`, `delete`, `do`, `else`, `enum`, `eval`, `export`, `extends`, `false`, `finally`, `for`, `function`, `if`, `implements`, `import`, `in`, `instanceof`, `interface`, `let`, `new`, `null`, `package`, `private`, `protected`, `public`, `return`, `static`, `super`, `switch`, `this`, `throw`, `true`, `try`, `typeof`, `var`, `void`, `while`, `with`, `yield`. These are all words that cannot be used without restrictions as names in the strict mode of ECMAScript 2020.

Strings

Strings are of the form "double-quote-characters", where double-quote-characters is a possibly empty sequence of characters without the character " and without the newline character, of the form 'single-quote-characters', where single-quote-characters is a possibly empty sequence of characters without the character ' and without the newline character, and of the form `backquote-characters`, where backquote-characters is a possibly empty sequence of characters without the character `. Note that newline characters are allowed as backquote-characters.
The following characters can be represented in strings as given:

- horizontal tab: \t
- vertical tab: \v
- nul char: \0
- backspace: \b
- form feed: \f
- newline: \n
- carriage return: \r
- single quote: \'
- double quote: \"
- backslash: \\

Unicode characters can be used in strings using \u followed by the hexadecimal representation of the unicode character, for example '\uD83D\uDC04'.

Comments

In Source, any sequence of characters between "/*" and the next "*/" is ignored.
After "//" any characters until the next newline character is ignored.

3 Dynamic Type Checking

Expressions evaluate to numbers, boolean values, strings or function values. Implementations of Source generate error messages when unexpected values are used as follows.
Only function values can be applied using the syntax:

expression ::= name (expressions)

For compound functions, implementations need to check that the number of expressions matches the number of parameters.
The following table specifies what arguments Source's operators take and what results they return. Implementations need to check the types of arguments and generate an error message when the types do not match.

operator	argument 1	argument 2	result
+	number	number	number
+	string	string	string
-	number	number	number
*	number	number	number
/	number	number	number
%	number	number	number
===	number	number	bool
===	string	string	bool
!==	number	number	bool
!==	string	string	bool
>	number	number	bool
>	string	string	bool
<	number	number	bool
<	string	string	bool
>=	number	number	bool
>=	string	string	bool
<=	number	number	bool
<=	string	string	bool
&&	bool	any	any
||	bool	any	any
!	bool		bool
-	number		number

Preceding `?` and following `if`, Source only allows boolean expressions.

4 Standard Libraries

The following libraries are always available in this language.

MISC Library

The following names are provided by the MISC library:

- `get_time()`: primitive, returns number of milliseconds elapsed since January 1, 1970 00:00:00 UTC
- `parse_int(s, i)`: primitive, interprets the string `s` as an integer, using the positive integer `i` as radix, and returns the respective value, see ECMAScript Specification, Section 18.2.5.
- `undefined`, `NaN`, `Infinity`: primitive, refer to JavaScript's undefined, NaN ("Not a Number") and Infinity values, respectively.
- `is_boolean(x)`, `is_number(x)`, `is_string(x)`, `is_undefined(x)`, `is_function(x)`: primitive, returns `true` if the type of `x` matches the function name and `false` if it does not. Following JavaScript, we specify that `is_number` returns `true` for `NaN` and `Infinity`.
- `prompt(s)`: primitive, pops up a window that displays the string `s`, provides an input line for the user to enter a text, a "Cancel" button and an "OK" button. The call of `prompt` suspends execution of the program until one of the two buttons is pressed. If the "OK" button is pressed, `prompt` returns the entered text as a string. If the "Cancel" button is pressed, `prompt` returns a non-string value.
- `display(x)`: primitive, displays the value `x` in the console[9]; returns the argument `a`.
- `display(x, s)`: primitive, displays the string `s`, followed by a space character, followed by the value `x` in the console[9]; returns the argument `x`.
- `error(x)`: primitive, displays the value `x` in the console[9] with error flag. The evaluation of any call of `error` aborts the running program immediately.

[9]The notation used for the display of values is consistent with JSON, but also displays `undefined` and function objects.

- `error(x, s)`: primitive, displays the string `s`, followed by a space character, followed by the value `x` in the console[9] with error flag. The evaluation of any call of `error` aborts the running program immediately.
- `stringify(x)`: primitive, returns a string that represents[9] the value `x`.

All library functions can be assumed to run in $O(1)$ time, except `display`, `error` and `stringify`, which run in $O(n)$ time, where n is the size (number of components such as pairs) of their first argument.

MATH Library

The following names are provided by the MATH library:

- `math_`name, where name is any name specified in the JavaScript `Math` library, see ECMAScript Specification, Section 20.2. Examples:
  - `math_PI`: primitive, refers to the mathematical constant π,
  - `math_sqrt(n)`: primitive, returns the square root of the number `n`.

All math functions can be assumed to run in $O(1)$ time and are considered primitive. All math functions expect numbers as arguments and return numbers. We don't specify the behavior of a math function when some arguments are not numbers.

Deviations from JavaScript

We intend the Source language to be a conservative extension of JavaScript: Every correct Source program should behave exactly the same using a Source implementation, as it does using a JavaScript implementation. We assume, of course, that suitable libraries are used by the JavaScript implementation, to account for the predefined names of each Source language. This section lists some exceptions where we think a Source implementation should be allowed to deviate from the JavaScript specification, for the sake of internal consistency and esthetics.

Evaluation result of programs: JavaScript statically distinguishes between value-producing and non-value-producing statements. All declarations are non-value-producing, and all expression statements, conditional statements and assignments are value-producing. A block is value-producing if its body statement is value-producing, and then its value is the value of its body statement. A sequence is value-producing if any of its component statements is value-producing, and then its value is the value of its last value-producing component statement. The value of an expression statement is the value of the expression. The value of a conditional statement is the value of the branch that gets executed, or the value `undefined` if that branch is not value-producing. The value of an assignment is the value of the expression to the right of its = sign. Finally, if the whole program is not value-producing, its value is the value `undefined`.

Example 1:

```
1;
{
  // empty block
}
```

The result of evaluating this program in JavaScript is 1.

Example 2:

```
1;
{
  if (true) {} else {}
}
```

The result of evaluating this program in JavaScript is `undefined`.

Implementations of Source are currently allowed to opt for a simpler scheme.

Hoisting of function declarations: In JavaScript, function declarations are "hoisted" (automagically moved) to the beginning of the block in which they appear. This means that applications of functions that are declared with function declaration statements never fail because the name is not yet assigned to their function value. The specification of Source does not include this hoisting; in Source, function declaration can be seen as syntactic sugar for constant declaration and lambda expression. As a consequence, application of functions declared with function declaration may fail in Source if the name that appears as function expression is not yet assigned to the function value it is supposed to refer to.